FEB'24: Fourier-mode analysis of function fit and its smoothing/regularization.
MAR'24: Slight change: write smoothing coefficient $\Delta x^2 \cdot d$ with d some dimensionless smoothing parameter as d_{art} of dimension $[\mathrm{m}^2]$ that scales with Δx^2.
This makes more sense: there should **not** be a (visible) discretization-dependent parameter in the easy problem!!!

In DEC'22, I wrote in the MAPLE worksheet **funcfit&adaptivesmoothing** from JUL'17 and later, that the following was to be included:
- F-mode analysis of 2nd-order piecewise linear and 4th-order piecewise cubic (one parameter) funcfit
- F-mode analysis of 2nd-order and 4th-order (dedicated? 2nd-order filter applied twice?) low-pass filter
- piecewise constant cell-based approximation, for comparison; extend to cell-based piecewise linear
- Forget about piecewise parabolic?
- 2D?

Let's put this F-mode stuff in a **separate** MAPLE worksheet, i.e., in this one. **Only** the 1D piecewise linear stuff, for the moment.

```
> restart;
  with(plots):
```

The **easy** equation to be solved for realization of smoothed/regularized function fit: $u^{\text{tilde}} - \frac{\partial}{\partial x}\left(d_{\text{art}} \cdot \frac{\partial u^{\text{tilde}}}{\partial x}\right) = u_{\text{giv}}$, with smoothing/regularization coefficient $d_{\text{art}} \sim \Delta x^2$. For the F-mode analyse we take grid size Δx and smoothing coefficient d_{art} constant. For convenience, we write $d_{\text{art}} = \Delta x^2 \cdot d$, with d a dimensionless smoothing parameter.

F-mode transformation of the ratio $u^{\text{tilde}} / u_{\text{giv}}$.

```
> Rtilde := 1 / ( 1 + d * kDx^2 );
  Rtilde := unapply( Rtilde, d ):
```

$$Rtilde := \frac{1}{d\, kDx^2 + 1}$$

NOTE: in the limit of $k \cdot \Delta x \to \infty$, R^{tilde} behaves as $1 / \left(d \cdot (k \cdot \Delta x)^2\right)$. Cut-off frequency determined by $1 = 1 / \left(d \cdot (k \cdot \Delta x)^2\right)$, hence filter parameter $d = 1 / (k \cdot \Delta x)^2$ gives cut-off frequency $k \cdot \Delta x$, or cut-off frequency $k \cdot \Delta x = \sqrt{1/d}$ is obtained with filter parameter d.

FVE discretization of smoothed/regularized function fit for **constant** Δx and **constant** d:

$$(1/8 - d) \cdot u^{\text{bar}}_{i-1} + (3/4 + 2 \cdot d) \cdot u^{\text{bar}}_{i} + (1/8 - d) \cdot u^{\text{bar}}_{i+1} = \frac{1}{\Delta x} \cdot \int_{x_{i-1/2}}^{x_{i+1/2}} u_{\text{giv}}\, \mathrm{d}x .$$

F-mode transformation of the $u^{\text{bar}} / u_{\text{giv}}$.

```
> Rbar := int( exp(I*kDx*i), i=-1/2..1/2 ) / ( 3/4 + 2*d + 2*(1/8-d)*cos(kDx) );
  Rbar := unapply( simplify(Rbar), d ):
```

$$Rbar := \frac{-I\left(e^{\frac{I}{2}kDx} - e^{-\frac{I}{2}kDx}\right)}{kDx\left(\frac{3}{4} + 2d + 2\left(\frac{1}{8} - d\right)\cos(kDx)\right)}$$

F-mode analysis of the error in an FVE function fit (exclude error due to low-pass filter, i.e., substitute $d=0$).

```
> errorFVEgridpoint := abs( Rbar(0) - 1 );
```

$$errorFVEgridpoint := \left|\frac{8\sin\left(\frac{kDx}{2}\right)}{kDx\left(-3 - \cos(kDx)\right)} + 1\right|$$

This error is for a F-mode fit of the numerical solution through the grid points, which is not the one closest to a piecewise linear function.
So let's also consider the F-mode FVE function fit error with the numerical F-mode through the cell centers.

```
> errorFVEcellcenter := Rbar(0)*cos(kDx/2) - 1;
  errorFVEcellcenter_lowestorder := series( errorFVEcellcenter, kDx, 4 );
  errorFVEcellcenter := abs( errorFVEcellcenter ):
```

$$errorFVEcellcenter := -\frac{8\sin\left(\frac{kDx}{2}\right)\cos\left(\frac{kDx}{2}\right)}{kDx\left(-3 - \cos(kDx)\right)} - 1$$

$$errorFVEcellcenter\_lowestorder := -\frac{1}{24}kDx^2 + O\left(kDx^4\right)$$

The best possible F-mode fit to a piecewise linear function approximation is a weighted mix between a fit through the grid points and a fit through the cell centers.
NOTE: we integrate the error, not the absolute value of the error, i.e., error canceling may occur. The result may be an optimistically (and incorrectly!) small error.

```
> errorFVEoptimal := abs( Rbar(0)*(1/3+2/3*cos(kDx/2)) - 1 );
```

$$errorFVEoptimal := \left|\frac{8\sin\left(\frac{kDx}{2}\right)\left(\frac{1}{3} + \frac{2\cos\left(\frac{kDx}{2}\right)}{3}\right)}{kDx\left(-3 - \cos(kDx)\right)} + 1\right|$$

For comparison: let's consider a piecewise linear function fit obtained from integration over cells (BOX-like).

```
> aux := int( exp(I*kDx*i), i=0..1 ) / ( 1/2 + exp(I*kDx)/2 ):
  # aux := simplify( aux );
  errorBOXlike := abs( aux - 1 );
```

$$errorBOXlike := \left|\frac{I\left(e^{I\,kDx} - 1\right)}{kDx\left(\frac{1}{2} + \frac{e^{I\,kDx}}{2}\right)} + 1\right|$$

Also for comparison: let's consider a piecewise linear function fit obtained from exact value at grid points (FD-like).

```
> assume( kDx > 0 ):
```

```
aux := int( exp(I*kDx*i) / ((1-i) + i*exp(I*kDx)), i=0..1 ):
# aux := simplify( aux );
errorFDlike := abs( aux - 1 );
```

$$errorFDlike := \left| \int_0^1 \frac{e^{I\, kDx\sim\, i}}{1 - i + i\, e^{I\, kDx\sim}}\, di - 1 \right|$$

Plot all of the above.

```
> loglogplot( [ Rtilde(2), Rbar(2), Rtilde(4), Rbar(4),
                errorFVEgridpoint, errorFVEcellcenter, errorFVEoptimal, errorBOXlike, errorFDlike ], kDx =
  Pi/100..Pi, 0.001..1, thickness=2,
              color=["Green", "Blue", "Green", "Blue", "Black", "Black", "Black", "Red", "Red"],
              linestyle=[3,3,1,1, 1,3,2,3,2], legendstyle=[location=right], size=[600,350],
              legend=["Rtilde(d=2)", "Rbar(d=2)", "Rtilde(d=4)", "Rbar(d=4)",
                      "errorFVEgridpoint", "errorFVEcellcenter", "errorFVEoptimal", "errorBOXlike",
  "errorFDlike"],
    labels=[typeset("\t        ",k*Dx)," attenuation,\n  discr.error\n\n\n\n\n\n\n\n\n"],
    tickmarks=[ [Pi/100=typeset(Pi,"/100"),Pi/50=typeset(Pi,"/50"),Pi/20=typeset(Pi,"/20"),
                 Pi/10=typeset(Pi,"/10"),Pi/5=typeset(Pi,"/5"),Pi/2=typeset(Pi,"/2"),Pi=typeset(Pi)],
                [.001="0.001",.002="0.002",.005="0.005",.01="0.01",.02="0.02",.05="0.05",.1="0.1",.2="0.2",
  .5="0.5",1="1.0"] ]
   );
```